

Wiedermal Excel Export



Stefan Glück

Wiedermal Excel Export 12.12.2015 08:13

Hallo,

ich hab wieder mal eine Frage zum Excel Export.
viele meiner Excel exportiere ich mit dieser Methode:

<http://www.qlikblog.at/971/qliktip-32-exporting-multiple-objects-single-excel-document/>

vor allem Wegen der Möglichkeit mehrere ExcelSheets zu erstellen.

jetzt habe ich aber das Problem Pivot Tabellen exportieren zu müssen, und da werden ja "doppelte Dimensionen" ausgeblendet. die CopytoClipboard Funktion belässt das so. Der normale Excel Export hingegen befüllt jede Zeile.

Gibt es eine Möglichkeit diese Zeilen im Pivot zu befüllen?

Ein Tabellendiagramm ist wegen der Dimensionssummen keine Alternative.

eine Möglichkeit für mich wäre 3 Excel zu erstellen und diese dann zu einem Zusammenzufassen. Allerdings habe ich keine Idee wie ich das anfangen könnte.

danke, mfg Stefan



Marcus Sommer 5.467 Einträge seit 23.03.2010

Re: Wiedermal Excel Export 12.12.2015 09:07

Hallo Stefan,

die Zielzeile gibt man ja generell vor und wenn man jetzt auch noch die Zeilen der Pivot zählt, könnte man über diesen Bereich eine vba-Schleife laufen lassen, die die Zeilen/Spalten drüber/drunter/daneben checkt und dann entsprechend ausfüllt oder zur nächsten Zeile geht. Das ist nicht super kompliziert, wird aber einige Zeit kosten, bis man das fehlerfrei implementiert hat.

Ich würde es aber anders versuchen und diese Pivots nicht kopieren sondern wirklich exportieren, danach diese Temp-Datei wieder öffnen und die Pivot in die eigentliche Excel rüberkopieren und das auch in die Hauptroutine integrieren.

Hier mal ein Beispiel für das Exportieren:

Wiedermal Excel Export

```
set obj = ActiveDocument.GetSheetObject("CH05")  
obj.ExportEx "Your Path\YourFile.xls",5
```

Im Forum gibt es eine Vielzahl von solchen/ähnlichen Beispielen.

Schöne Grüße

Marcus



Christian Wolf

Re: Wiedermal Excel Export 14.12.2015 07:55

Hallo Stefan,

wenn eine Pivot Tabelle über die Zwischenablage nach Excel kopiert wird, enthalten die Dimensionsspalten im Excel verbundene Zellen.

Mit den folgenden Makros können diese Verbunde aufgelöst werden:

```
Sub SearchAreas(ByRef objExcelApp, startCells, numberOfRows)  
    aStartCells = Split(startCells, ",")  
    For i = LBound(aStartCells) To UBound(aStartCells)  
        addr = GetColRow(aStartCells(i))  
        lastRow = numberOfRows+addr(1)  
        Do Until addr(1) > lastRow  
            addr(1) = ResolveArea(objExcelApp, addr)  
        Loop  
    Next  
End Sub  
  
Function ResolveArea(ByRef objExcelApp, startCell)  
    Set range1 = objExcelApp.range(startCell(0) & startCell(1))  
    addr = range1.MergeArea.Address  
  
    If InStr(addr, ":") > 0 Then  
        vValue = range1.Cells(1).Value  
        range1.Cells(1).MergeCells = False  
  
        Set range2 = objExcelApp.range(addr)  
        For Each cell In range2  
            cell.Value = vValue  
        Next  
  
        tmp = GetColRows(addr)  
        ret = tmp(3) + 1  
    Else
```

Wiedermal Excel Export

```
        ret = startCell(1) + 1
    End If

    ResolveArea = ret
End Function

'Hilfsfunktionen
Function GetColRow(cell)
    Dim out(2)
    cell = Replace(cell, "$", "")
    out(0) = ""
    out(1) = ""
    For l = 1 To Len(cell)
        s = Mid(cell, l, 1)
        If IsNumeric(s) Then
            out(1) = out(1) & s
        Else
            out(0) = out(0) & s
        End If
    Next

    out(1) = CInt(out(1))
    GetColRow = out
End Function

Function GetColRows(range)
    Dim out(4)
    aCell = Split(range, ":")
    tmp = GetColRow(aCell(0))
    out(0) = tmp(0)
    out(1) = tmp(1)
    tmp = GetColRow(aCell(1))
    out(2) = tmp(0)
    out(3) = tmp(1)
    GetColRows = out
End Function
```

In deinem Export Makro fügst du dann unmittelbar nach der Zeile `objExcelDoc.Sheets(sheetName).Paste` zum Beispiel für eine Pivotabelle die bei A1 eingefügt wird, eine Kopfzeile hat und 3 Dimensionsspalten die folgenden Zeilen ein:

```
rows = objExcelApp.Selection.Rows.Count
SearchAreas objExcelApp, "A2,B2,C2", rows
```

Das Makro sucht in diesem Beispiel in den Spalten A,B und C, jeweils beginnend ab Zeile 2 über den eingefügten Bereich nach Zellverbunden und löst diese auf.

Grüße

Wiedermal Excel Export

Christian



Stefan Glück

Re: Wiedermal Excel Export 14.12.2015 07:59

danke für eure Tipps, da hab ich mal richtig viel zum Testen



Stefan Glück

Re: Wiedermal Excel Export 14.12.2015 11:19

Habs getestet. funktioniert einwandfrei für meine Zwecke. danke. dadurch kann ich mehr Formatierungen im Qlikview direkt machen und erspare mir eine Menge Macrolaufzeit!

danke